CHAPTER 1

LOGICAL FUNCTIONS

INFOCUS

Logical functions are used in worksheets to test whether a situation is true or false. Depending on the result of that test, you can then choose to perform an action such as display information, perform different calculations or perform further tests.

In this session you will:

- ✓ gain an understanding of logical functions
- ✓ learn how to use ***IF*** with text
- ✓ learn how to use ***IF*** with numbers
- ✓ learn how to nest ***IF*** functions
- ✓ learn how to use ***IFERROR***
- ✓ learn how to use ***TRUE*** and ***FALSE***
- ✓ learn how to use the ***AND*** function
- ✓ learn how to use the ***OR*** function
- ✓ learn how to use the ***NOT*** function.

UNDERSTANDING LOGICAL FUNCTIONS

Logical functions provide decision-making tools for information in a spreadsheet. They allow you to look at the contents of a cell or to perform a calculation, and then test that result against a required figure or value. You can then use the ***IF*** logical function to determine which calculation to perform or action to take depending on the outcome of the test. Here are some examples.

The IF Function

The ***IF*** function is the key logical function used for decision making. It takes the format:

=IF(condition, true, false)

For example, you could use the following formula:

=IF(B2 > 400, “High”, “Low”) where,

B2 > 400	is the ***condition*** being tested (this could be translated as “Is the value in cell B2 greater than 400?”)
“High”	is the text to display if B2 is greater than 400 (the result of the test is ***yes*** or ***TRUE***)
“Low”	is the text to display if B2 is less than or equal to 400 (the result of the test is ***no*** or ***FALSE***)

The AND Function

The ***AND*** function is used to compare more than one condition. It returns TRUE only if all of the conditions are met, and takes the format:

=AND(condition1, condition2,…)

For example, you could use the following formula:

=AND(B2 > 400, C2 < 300) where,

B2 > 400	is the first condition being tested
C2 < 300	is the second condition being tested

This will only return the result ***TRUE*** if the value in cell B2 is greater than 400 ***and*** the value in cell C2 is less than 300. In all other situations, the result will be ***FALSE***.

The OR Function

The ***OR*** function is also used to compare more than one condition. It returns TRUE if any of the conditions are met, and takes the format:

=OR(condition1, condition2,…)

For example, you could use the following formula:

=OR(B2 > 400, C2 < 300) where,

B2 > 400	is the first condition being tested
C2 < 300	is the second condition being tested

This will return the result ***TRUE*** if either the value in cell B2 is greater than 400 ***or*** the value in cell C2 is less than 300. The result will be ***FALSE*** only if none of the conditions are met.

USING IF WITH TEXT

The ***IF*** function can be used to display different information depending on the outcome of a *conditional* test. The resulting text will appear in the cell where the formula containing the ***IF*** function resides. In this example, the ***IF*** function is used to indicate whether adjacent sales figures meet or exceed a specified target. This makes identifying successful sales people far easier.

Try This Yourself:

Open File *Before starting this exercise you MUST open the file Logical Functions_1.xlsx...*

1. Click on the ***IF Function*** worksheet tab
2. Click in cell ***D7***, then type:

   =IF(C7>$E$2,"Exceeded Target","Below Target")
3. Press Enter to complete the formula
4. Click in cell ***D7***, then double-click on the fill handle to copy the formula down the column

   Notice that the result for Jerry Hancock is Below Target even though he achieved 34,000. This is because the condition requires the sales to be greater than the target. We'll change it to greater than or equal to the target...
5. Click in cell ***D7***, then click in the ***Formula Bar*** immediately to the right of **>**
6. Type **=**, then press Enter
7. Click in cell ***D7***, then double -click on the fill handle to copy the formula down the column

2

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	=IF(C7>$E$2,"Exceeded Target","Below Target")			
8	Mark	Daniels	25,000				
9	Maureen	Grayson	27,800				
10	Jerry	Hancock	34,000				
11	Brian	Houson	18,350				
12	Helen	Kai	12,500				
13	Norris	Maunga	75,880				
14	Alex	Nguyen	43,778				
15	Kate	Rualowy	23,400				
16							

4

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	Exceeded Target			
8	Mark	Daniels	25,000	Below Target			
9	Maureen	Grayson	27,800	Below Target			
10	Jerry	Hancock	34,000	Below Target			
11	Brian	Houson	18,350	Below Target			
12	Helen	Kai	12,500	Below Target			
13	Norris	Maunga	75,880	Exceeded Target			
14	Alex	Nguyen	43,778	Exceeded Target			
15	Kate	Rualowy	23,400	Below Target			
16							

7

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	Exceeded Target			
8	Mark	Daniels	25,000	Below Target			
9	Maureen	Grayson	27,800	Below Target			
10	Jerry	Hancock	34,000	Exceeded Target			
11	Brian	Houson	18,350	Below Target			
12	Helen	Kai	12,500	Below Target			
13	Norris	Maunga	75,880	Exceeded Target			
14	Alex	Nguyen	43,778	Exceeded Target			
15	Kate	Rualowy	23,400	Below Target			
16							

For Your Reference…

To ***use*** the ***IF function*** to ***create decision making***:

=IF(test, value_if_true, value_if_false)

This function performs a **test**, then if the result is true, uses the entry in the position **true**. If the result is not true, the entry for **false** is used.

Handy to Know…

- If you only want text to appear if the result is true, you can enter **""** (two double quotes) in the position for ***false***. For example, ***=IF(C7>=$E$2,"Exceeded Target","")*** will only display text if the target was met or exceeded.

USING IF WITH NUMBERS

One of the most common uses of the ***IF*** function is to perform numerical computations based on the outcome of the condition test. This is achieved by putting formulas that would normally be used to calculate values in place of the ***true*** and ***false*** components in the function. You can also use this structure to show a specific value according to the result of the condition test.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file Logical Functions_2.xlsx...

1. Ensure the ***IF Function*** worksheet tab is selected, then click in cell ***E7***
2. Type:
   =IF(C7>=$E$2,(C7-$E$2)*$E$3,0)
   This formula tests to see if the monthly sales figure is greater than or equal to the target. If it is, then the difference between the monthly sales and target is calculated and this difference is multiplied by the 5% commission. If the monthly sales isn't greater than or equal to the target, a zero commission is applicable...
3. Press Enter to complete the formula
4. Click in cell ***E7***, then double-click on the fill handle to copy the formula down

   We formatted column E before you opened the workbook – that's why the dash symbol appears instead of zero where no commission is to be paid

2

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	Exceeded Target	=IF(C7>=$E$2,(C7-$E$2)*$E$3,0)		
8	Mark	Daniels	25,000	Below Target			
9	Maureen	Grayson	27,800	Below Target			
10	Jerry	Hancock	34,000	Exceeded Target			
11	Brian	Houson	18,350	Below Target			
12	Helen	Kai	12,500	Below Target			
13	Norris	Maunga	75,880	Exceeded Target			
14	Alex	Nguyen	43,778	Exceeded Target			
15	Kate	Rualowy	23,400	Below Target			
16							

3

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	Exceeded Target	550		
8	Mark	Daniels	25,000	Below Target			
9	Maureen	Grayson	27,800	Below Target			
10	Jerry	Hancock	34,000	Exceeded Target			
11	Brian	Houson	18,350	Below Target			
12	Helen	Kai	12,500	Below Target			
13	Norris	Maunga	75,880	Exceeded Target			
14	Alex	Nguyen	43,778	Exceeded Target			
15	Kate	Rualowy	23,400	Below Target			
16							

4

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	Exceeded Target	550		
8	Mark	Daniels	25,000	Below Target	-		
9	Maureen	Grayson	27,800	Below Target	-		
10	Jerry	Hancock	34,000	Exceeded Target	-		
11	Brian	Houson	18,350	Below Target	-		
12	Helen	Kai	12,500	Below Target	-		
13	Norris	Maunga	75,880	Exceeded Target	2,094		
14	Alex	Nguyen	43,778	Exceeded Target	489		
15	Kate	Rualowy	23,400	Below Target	-		
16							

For Your Reference…

To ***use*** the ***IF function*** to ***create decision making*** with ***numeric values***:

=IF(test, value_if_true, value_if_false)

The true-value and the false-value both need to be numeric here. They could be a value, a cell reference, or even another (nested) formula.

Handy to Know…

- The ***true_value*** and ***false_value*** in an ***IF*** function can be mixed with one text and the other a numeric.

NESTING IF FUNCTIONS

If you need to make more than one decision before calculating an answer, you can *nest* or embed an ***IF*** function inside another ***IF*** function. For example, you can use an ***IF*** function in place of the ***true*** component of the ***IF*** function. If the result of the first condition test is true, the second condition will be tested. This structure provides for three alternative outcomes instead of two.

Try This Yourself:

Same File *Continue using the previous file with this exercise, or open the file Logical Functions_3.xlsx...*

1. Double-click in cell ***E7*** to display the formula
2. Click after the first equal sign, type **IF(C7>=(2*$E$2),** then press Alt + Enter to create a new line

   You don't have to put a new line in the Formula Bar for the formula to work but it does make it easier to work with complex formulas…
3. Type **(C7-$E$2)*(2*$E$3),** then press Alt + Enter to create a new line
4. Click immediately after the first comma on the last line, then press Alt + Enter to create a new line
5. Again, click immediately after the first comma on the last line, then press Alt + Enter

   Your formula is now divided into components. Let's complete the formula…
6. Press End to move to the end of the formula, then type **)**
7. Press Enter, click in cell ***E7*** then click and drag the fill handle down to cell ***E15***

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	Exceeded Target	=IF(C7>=(2*$E$2),		
8	Mark	Daniels	25,000	Below Target	(C7-$E$2)*(2*$E$3),		
9	Maureen	Grayson	27,800	Below Target	IF(C7>=$E$2,		
10	Jerry	Hancock	34,000	Exceeded Target	(C7-$E$2)*$E$3,		
11	Brian	Houson	18,350	Below Target	0)		
12	Helen	Kai	12,500	Below Target	IF(logical_test, [value_if_true], [value_if		
13	Norris	Maunga	75,880	Exceeded Target	2,094		
14	Alex	Nguyen	43,778	Exceeded Target	489		
15	Kate	Rualowy	23,400	Below Target	-		
16							

5

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Status	Commission		
6							
7	Janet	Costas	45,000	Exceeded Target	550		
8	Mark	Daniels	25,000	Below Target	-		
9	Maureen	Grayson	27,800	Below Target	-		
10	Jerry	Hancock	34,000	Exceeded Target	-		
11	Brian	Houson	18,350	Below Target	-		
12	Helen	Kai	12,500	Below Target	-		
13	Norris	Maunga	75,880	Exceeded Target	4,188		
14	Alex	Nguyen	43,778	Exceeded Target	489		
15	Kate	Rualowy	23,400	Below Target	-		
16							

7

For Your Reference…

To ***nest*** one ***IF inside another***:

1. Double-click on the formula cell
2. Click at the desired location in the formula either in the ***Formula bar*** or in the cell being edited
3. Type the additional requirements

Handy to Know…

- When you create nested formulas, Excel will colour-code the paired brackets to make it easier to see what you are doing. The outside brackets are coloured black.
- You can nest any function within another function, but plan carefully.

USING IFERROR

IFERROR is used to trap errors that may occur as the result of a calculation and then display alternative text or values. For example, if you divide a number by zero, Excel will normally return the message ***#DIV/0!*** which can be a bit alarming for novice users. ***IFERROR*** tests a calculation to see if it works and, if so, performs the calculation. If not, it displays an alternative.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file Logical Functions_4.xlsx...

1. Click on the ***IFERROR Function*** worksheet tab, then click in cell ***E7***
2. Type the following

   =IFERROR(C7/D7, "First Year")
3. Press Enter
4. Click in cell ***E7***, then double-click on the fill handle to copy the formula down the column

   Instead of giving an error where the divisor is zero, Excel displays the text "First Year" in the cell

2

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Average Annual Sales						
3							
4							
5	Agent		Total Sales	Years as Agent	Average Annual Sales		
6							
7	Janet	Costas	2,578,015	2	=IFERROR(C7/D7, "First Year")		
8	Mark	Daniels	4,875,485	4			
9	Maureen	Grayson	2,978,450	3			
10	Jerry	Hancock	7,586,204	6			
11	Brian	Houson	1,083,650	0			
12	Helen	Kai	1,284,500	0			
13	Norris	Maunga	7,658,900	8			
14	Alex	Nguyen	4,357,859	5			
15	Kate	Rualowy	2,487,652	3			
16							

4

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Average Annual Sales						
3							
4							
5	Agent		Total Sales	Years as Agent	Average Annual Sales		
6							
7	Janet	Costas	2,578,015	2	1,289,008		
8	Mark	Daniels	4,875,485	4	1,218,871		
9	Maureen	Grayson	2,978,450	3	992,817		
10	Jerry	Hancock	7,586,204	6	1,264,367		
11	Brian	Houson	1,083,650	0	First Year		
12	Helen	Kai	1,284,500	0	First Year		
13	Norris	Maunga	7,658,900	8	957,363		
14	Alex	Nguyen	4,357,859	5	871,572		
15	Kate	Rualowy	2,487,652	3	829,217		
16							

For Your Reference…

IFERROR(calculation, error_value)

This function performs the ***calculation*** and if there are no errors, displays the result of the calculation. If an error does occur, it displays the ***error_value***.

Handy to Know…

- When using ***IFERROR*** you can use text as the entry to be displayed if an error is located, but you could just as easily display nothing using **""** (two double quotes) or perform an alternative calculation.

USING TRUE AND FALSE

TRUE and ***FALSE*** are logical values. The result of a logical test is either true or false and Excel allows you to enter these values in cells or test for them in functions. TRUE and FALSE can be entered as ***values***, which are TRUE and FALSE, or as ***formulas*** with no parameters, which are =TRUE() and =FALSE(). The value TRUE and the formula =TRUE() are treated as identical by Excel.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file Logical Functions_5.xlsx...

1. Click on the ***AND Function*** worksheet tab, then click in cell ***D7***
2. Type the **TRUE** and **FALSE** entries in the column as shown
3. Click in cell ***E7***, then type **=IF(C7>=$E$2, IF(D7=TRUE, (C7-$E$2)*$E$3,0),0)**
4. Press Enter
5. Click in cell ***E7***, then double-click on the fill handle to copy the formula down the column

	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	On Staff	Commission
6					
7	Janet	Costas	45,000	TRUE	
8	Mark	Daniels	25,000	TRUE	
9	Maureen	Grayson	27,800	FALSE	
10	Jerry	Hancock	34,000	FALSE	
11	Brian	Houson	18,350	FALSE	
12	Helen	Kai	12,500	TRUE	
13	Norris	Maunga	75,880	TRUE	
14	Alex	Nguyen	43,778	FALSE	
15	Kate	Rualowy	23,400	FALSE	
16					

2

	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	On Staff	Commission
6					
7	Janet	Costas	45,000	TRUE	=IF(C7>=$E$2, IF(D7=TRUE, (C7-$E$2)*$E$3,0),0)
8	Mark	Daniels	25,000	TRUE	
9	Maureen	Grayson	27,800	FALSE	
10	Jerry	Hancock	34,000	FALSE	
11	Brian	Houson	18,350	FALSE	
12	Helen	Kai	12,500	TRUE	
13	Norris	Maunga	75,880	TRUE	
14	Alex	Nguyen	43,778	FALSE	
15	Kate	Rualowy	23,400	FALSE	
16					

3

	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	On Staff	Commission
6					
7	Janet	Costas	45,000	TRUE	550
8	Mark	Daniels	25,000	TRUE	-
9	Maureen	Grayson	27,800	FALSE	-
10	Jerry	Hancock	34,000	FALSE	-
11	Brian	Houson	18,350	FALSE	-
12	Helen	Kai	12,500	TRUE	-
13	Norris	Maunga	75,880	TRUE	2,094
14	Alex	Nguyen	43,778	FALSE	-
15	Kate	Rualowy	23,400	FALSE	-
16					

5

For Your Reference…

TRUE

The logical value TRUE

FALSE

The logical value FALSE

Handy to Know…

- ***TRUE*** is used to make formulas more readable. For instance, you could write the formula, **=IF(C7>=$E$2, IF(D7=TRUE, (C7-$E$2)*$E$3,0),0)** as **=IF(C7>=$E$2,IF(D7,(C7-$E$2)*$E$3,0),0)**, however including ***TRUE*** in the first example makes the formula easier to understand.

USING AND

The ***AND*** function is used to compare the results of more than one condition test. It ensures that a calculation will not be performed unless all of the specified conditions are met. In other words, the first and second and third (and so on) conditions must all be true before ***AND*** returns the value ***true***. This is ideal to use with the IF function to test for a collection of conditions.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file Logical Functions_6.xlsx...

1. Ensure the ***AND Function*** worksheet is selected, click in cell ***E7***, then type: **=AND(C7>=$E$2,D7=TRUE)**
2. Press Enter

   The result will be TRUE because both conditions are satisfied. Now to add the IF function...
3. Double-click in cell ***E7***, click after the first equal sign, then type **IF(**
4. Press End to move to the end of the formula, type **,** (comma), then press Alt + Enter to create a new line
5. Type **(C7-$E$2)*$E$3,0)**
6. Press Enter
7. Click in cell ***E7***, then double-click on the fill handle to copy the formula down the column

1

	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	On Staff	Commission
6					
7	Janet	Costas	45,000	TRUE	=AND(C7>=$E$2,D7=TRUE)
8	Mark	Daniels	25,000	TRUE	-
9	Maureen	Grayson	27,800	FALSE	-
10	Jerry	Hancock	34,000	FALSE	-
11	Brian	Houson	18,350	FALSE	-
12	Helen	Kai	12,500	TRUE	-
13	Norris	Maunga	75,880	TRUE	2,094
14	Alex	Nguyen	43,778	FALSE	-
15	Kate	Rualowy	23,400	FALSE	-
16					

5

	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	On Staff	Commission
6					
7	Janet	Costas	45,000	TRUE	=IF(AND(C7>=$E$2,D7=TRUE),
8	Mark	Daniels	25,000	TRUE	(C7-$E$2)*$E$3,0)
9	Maureen	Grayson	27,800	FALSE	-
10	Jerry	Hancock	34,000	FALSE	-
11	Brian	Houson	18,350	FALSE	-
12	Helen	Kai	12,500	TRUE	-
13	Norris	Maunga	75,880	TRUE	2,094
14	Alex	Nguyen	43,778	FALSE	-
15	Kate	Rualowy	23,400	FALSE	-
16					

7

	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	On Staff	Commission
6					
7	Janet	Costas	45,000	TRUE	550
8	Mark	Daniels	25,000	TRUE	-
9	Maureen	Grayson	27,800	FALSE	-
10	Jerry	Hancock	34,000	FALSE	-
11	Brian	Houson	18,350	FALSE	-
12	Helen	Kai	12,500	TRUE	-
13	Norris	Maunga	75,880	TRUE	2,094
14	Alex	Nguyen	43,778	FALSE	-
15	Kate	Rualowy	23,400	FALSE	-
16					

For Your Reference…

AND(logical1, logical2,…)

This function tests the logical value of each entry e.g. ***logical1***. If they are all true, it will return the value ***TRUE***. If any one of them is false, the function will return ***FALSE***.

Handy to Know…

- A condition in an ***AND*** function can simply be a reference to a cell holding a logical value (that is, TRUE or FALSE). For example, ***=AND(B2,C2)*** will return the value FALSE if cell B2 and/or cell C2 contain the text FALSE.

USING OR

The ***OR*** function is used to compare the results of more than one condition test. It will return the value ***TRUE*** if any of the condition tests return the value ***TRUE***. It will only return the value ***FALSE*** if all of the condition tests return ***FALSE***. The ***OR*** function is often used in conjunction with the ***IF*** function to test a collection of conditions, and is easier to work with than nested ***IF*** functions.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file Logical Functions_7.xlsx...

1. Click on the ***OR Function*** worksheet tab, then click in cell ***E7***
2. Type

   =OR(D7="Gold",D7="Silver")
3. Press Enter

   The result will be TRUE because both conditions are satisfied. Now to add the IF function…
4. Double-click in cell ***E7***, click after the first equal sign, then type **IF(**
5. Press End to move to the end of the formula, type **,** (comma), then press Alt + Enter to create a new line
6. Type **(C7-$E$2)*$E$3,0)**
7. Press Enter
8. Click in cell ***E7***, then double-click on the fill handle to copy the formula down the column

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Agent Classification	Commission		
6							
7	Janet	Costas	45,000	Gold	=OR(D7="Gold",D7="Silver")		
8	Mark	Daniels	25,000	Bronze			
9	Maureen	Grayson	27,800	Bronze			
10	Jerry	Hancock	34,000	Silver			
11	Brian	Houson	18,350	Bronze			
12	Helen	Kai	12,500	Bronze			
13	Norris	Maunga	75,880	Gold			
14	Alex	Nguyen	43,778	Gold			
15	Kate	Rualowy	23,400	Silver			
16							

2

	A	B	C	D	E	F	G
1	Alpheius Global Enterprises						
2	Agency Commissions			Target	34,000		
3				Commission	5%		
4							
5	Agent		Monthly Sales	Agent Classification	Commission		
6							
7	Janet	Costas	45,000	Gold	=IF(OR(D7="Gold",D7="Silver"),		
8	Mark	Daniels	25,000	Bronze	(C7-$E$2)*$E$3,0)		
9	Maureen	Grayson	27,800	Bronze			
10	Jerry	Hancock	34,000	Silver			
11	Brian	Houson	18,350	Bronze			
12	Helen	Kai	12,500	Bronze			
13	Norris	Maunga	75,880	Gold			
14	Alex	Nguyen	43,778	Gold			
15	Kate	Rualowy	23,400	Silver			
16							

6

For Your Reference…

OR(logical1, logical2,…)

This function tests the specified ***logical*** conditions or cell references. If any one of the conditions is true, it will return the value ***TRUE***. If all of them are false, the function will return ***FALSE***.

Handy to Know…

- If you want to find data that meets more than one condition you can use the ***AND*** function and the ***OR*** function together.

USING NOT

Sometimes the best way to get the result you need is to exclude the values that you don't want, rather than testing for the values that you do. The ***NOT*** function is perfect for this situation, returning the logical opposite of the condition test. If the condition test returns the value ***TRUE***, the ***NOT*** function will return the value ***FALSE***. This can also be used to great effect with the ***IF*** function.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file Logical Functions_8.xlsx...

1. Ensure the ***OR Function*** worksheet is selected, double-click in cell ***E7***, then click before ***OR***
2. Press Del twice to remove the ***OR*** function, then change the formula so it reads

   =IF(",C7-$E$2)*$E$3,0)
3. Add the ***NOT*** function so that the formula now reads

   =IF(NOT(D7="Bronze"),(C7-$E$2)*$E$3,0)

   This excludes any of the Bronze sales agents, therefore selecting Silver and Gold…
4. Press Enter
5. Click in cell ***E7***, then double-click on the fill handle to copy the formula down the column

   The results are the same, we've just tested the contents of the cells in a slightly different way



	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	Agent Classification	Commission
6					
7	Janet	Costas	45,000	Gold	=IF(",C7-$E$2)*$E$3,0)
8	Mark	Daniels	25,000	Bronze	
9	Maureen	Grayson	27,800	Bronze	-
10	Jerry	Hancock	34,000	Silver	-
11	Brian	Houson	18,350	Bronze	-
12	Helen	Kai	12,500	Bronze	-
13	Norris	Maunga	75,880	Gold	2,094
14	Alex	Nguyen	43,778	Gold	489
15	Kate	Rualowy	23,400	Silver	- 530
16					

IF(**logical_test**, [value_if_true], [value_if_false])

2



	A	B	C	D	E
1	Alpheius Global Enterprises				
2	Agency Commissions			Target	34,000
3				Commission	5%
4					
5	Agent		Monthly Sales	Agent Classification	Commission
6					
7	Janet	Costas	45,000	Gold	=IF(NOT(D7="Bronze"),(C7-$E$2)*$E$3,0)
8	Mark	Daniels	25,000	Bronze	
9	Maureen	Grayson	27,800	Bronze	-
10	Jerry	Hancock	34,000	Silver	-
11	Brian	Houson	18,350	Bronze	-
12	Helen	Kai	12,500	Bronze	-
13	Norris	Maunga	75,880	Gold	2,094
14	Alex	Nguyen	43,778	Gold	489
15	Kate	Rualowy	23,400	Silver	- 530
16					

IF(logical_test, **[value_if_true]**, [value_if_false])

3

For Your Reference…

NOT(logical)

This function tests the specified ***logical*** condition or contents of a cell. If the condition is true, it will return the value ***FALSE***. If the condition is false, the function will return ***TRUE***.

Handy to Know…

- If you want to pay commission to Gold and Silver agents only if they exceed the target (to avoid paying negative commission), you can use:

  =IF(AND(C7>=$E$2, NOT(D7="Bronze")),(C7-$E$2)*$E$3,0)

CHAPTER 2

LOOKUP FUNCTIONS

INFOCUS

Excel provides a number of functions that allow you to look up and extract data from a list or table. These are known as ***Lookup*** functions and they can be used for a variety of purposes, such as:

- returning the appropriate tax rate based on salary
- returning the data that is at, say, the second column, third row of a table
- returning the description, price and discount rate of an item, based on its code in the data inventory.

In this session you will:

- ✓ gain an understanding of data lookup functions
- ✓ learn how to use the ***CHOOSE*** function
- ✓ learn how to use ***VLOOKUP***
- ✓ learn how to use ***VLOOKUP*** for exact matches
- ✓ learn how to use ***HLOOKUP***
- ✓ learn how to use ***INDEX***
- ✓ learn how to use the ***MATCH*** function
- ✓ gain an understanding of reference functions
- ✓ learn how to use ***ROW*** and ***ROWS***
- ✓ learn how to use ***COLUMN*** and ***COLUMNS***
- ✓ learn how to use ***ADDRESS***
- ✓ learn how to use ***INDIRECT***
- ✓ learn how to use ***OFFSET***.

UNDERSTANDING DATA LOOKUP FUNCTIONS

Data lookup functions are used to retrieve data from a table. They generally require at least two pieces of information; what to look for and where to look for it. The *what to look for* part is often part of a table of information which can be referred to as a ***calculation area***. The *where to look for it* is known as a ***data table*** – a table in which a list of rates, figures, text or other items are held.

Data Area

The ***data area*** is often on a worksheet by itself, protecting it from accidentally being modified or deleted. It holds all of the possible values for the data. The values are laid out in a table format and they are listed in numerical or alphabetical order of the code that the lookup function will search for.

In this example, we have created the name ***Pay_Rates*** for the range ***B3:C7*** that holds the data. The resulting formula in the calculation area will be easier to understand.



Calculation Area

The ***calculation area*** is usually on a worksheet separate to the data area, unless you require the data values to be visible as well as the resulting calculations.

The calculation area uses a formula, such as ***VLOOKUP***, to find the correct data for each situation. In this example, the ***VLOOKUP*** function shown is comparing the value in cell ***C5*** with the values in the range ***Pay_Rates***. It then returns the value in the second column of the data table, determined by the ***2*** in the formula.